
XVERWEIS und XVERGLEICH 

Excelstammtisch vom 10.02.2020 

(zusammengefasst von René Martin) 

1. Die neue Funktion XVERWEIS 

In Excel für Office 365 in der Version 2001 gibt es nun zwei neue Funktionen: XVERWEIS und 
XVERGLEICH. Die Syntax von XVERWEIS sieht folgendermaßen aus: 

=XVERWEIS(Suchkriterium;Suchmatrix;Rückgabematrix; 
[wenn nicht gefunden];[Vergleichsmodus];[Suchmodus]) 

2. Ähnlich wie SVERWEIS … 

Die ersten drei notwendigen Parameter arbeiten ähnlich wie SVERWEIS: gesucht wird ein Wert 
in einer Spalte – zurückgegeben wird der Wert einer anderen Spalte: 

 

Sind die beiden Bereiche nicht gleich groß, ist eine Fehlermeldung (#Wert) die Folge. Sind die 
Bereiche „versetzt“, also: suche in A2:A46 und liefere den Wert aus A4:A48, dann ist das 
Ergebnis falsch.  

XVERWEIS bildet nicht nur SVERWEIS nach, sondern auch WVERWEIS – man kann auch 
zeilenweise suchen: 

Da in einer Spalte gesucht und in einer anderen Spalte gefunden wird, bietet sich eine 
„intelligente“ (formatierte/strukturierte/dynamische) Tabelle an: 

 

3. Ähnlich wie INDEX und VERGLEICH 

Da Suchbereich und Rückgabebereich unabhängig sind, ersetzt die Funktion XVERWEIS die 
beiden Funktionen INDEX und VERGLEICH: 

 

Hinweis 

Anders als beim SVERWEIS ist die Standardeinstellung: „suche exakt“: 

4. Der Vergleichsmodus 

Die Standardeinstellung von SVERWEIS: WAHR entspricht beim XVERWEIS dem 
Parameterwert -1 beim „Vergleichsmodus“: 

 

Ebenso wie bei der Funktion VERGLEICH kann der darunterliegende Wert zurückgegeben 
werden: 

 

Und schließlich kann der vierte Parameterwert 2 („Platzhalterzeichenübereinstimmung“) mit den 
beiden Platzhaltern „*“ und „?“ arbeiten – „*“ als Platzhalter für beliebige Zeichen (0 Mal, 1 Mal 
oder beliebige) „?“ als Platzhalter für genau ein Zeichen. 

Hinweis 

Die beiden Platzhalter sind aus dem Autofilter bekannt und werden hier genauso 
verwendet. Ebenso wie beim Autofilter unterscheidet XVERWEIS nicht zwischen Groß- 
und Kleinschreibung. 

5. Der Parameter wenn_nicht_gefunden 

Wird ein Wert nicht gefunden, weil er (bei den Parameterwerten 0 und 2) nicht vorhanden ist 
oder weil er (bei den Parameterwerten 1 und -1) außerhalb des Bereiches liegt, ist #NV der 
Fehler. Er kann abgefangen werden mit dem Parameter „wenn_nicht_gefunden“: 

 

Da SVERWEIS die Fehler #BEZUG und #NV liefern kann und diese mit WENNNV, 
beziehungsweise WENNFEHLER abgefangen werden können, so ist die Funktion WENNNV 
in der Funktion XVERWEIS bereits integriert. 

6. Der Parameter Suchmodus 

Der letzte Parameter „Suchmodus“ kann 1: von oben nach unten oder 2: von unten nach oben 
suchen und finden. Dies ist bei mehrfach auftretenden Werten interessant. 

  

Ein weiteres Beispiel: erster Wert und letzter Wert: 

Da XVERWEIS die beiden Funktionen SVERWEIS und WVERWEIS vereint, kann man ihn 
kombinieren: 

=XVERWEIS(N5;tbl_Kunden2[Nr]; 
XVERWEIS(O$4;tbl_Kunden2[#Kopfzeilen];tbl_Kunden2)) 

 

7. XVERWEIS als Array 

XVERWEIS liefert einen Wert oder mehrere Werte. Sie werden als Array angezeigt: 

=XVERWEIS(O2;tbl_Kunden3[Nr]; 
tbl_Kunden3[Vorname]:tbl_Kunden3[Ort];"nicht gefunden") 

 

Ebenso kann XVERWEIS als Datenbereich verwendet werden: 

=SUMME(XVERWEIS(E3;A2:A62;B2:B62):XVERWEIS(E5;A2:A62;B2:B62)) 

 

Und somit ist möglich Werte bis zu einem bestimmten Monat oder von dem ersten „x“ bis zum 
letzten „x“ zu summieren: 

=SUMME(XVERWEIS("x";B:B;A:A;;;1):XVERWEIS("x";B:B;A:A;;;-1)) 

 

Auch das funktioniert: 

=SUMME(B1:XVERWEIS(D2;A2:A13;B2:B13)) 

 

Und man kann auch mehrere Werte verketten und das Ergebnis suchen: 

=XVERWEIS(F15&"-"&TEXTKETTE(G15:J15);A2:A10;F2:F10) 

Oder umgekehrt: einen Wert in mehreren Spalten suchen: 

=XVERWEIS(F15;A2:A10&"-"&B2:B10;F2:F10) 

8. Dynamische Diagramme 

Mit den drei Funktionen BEREICH.VERSCHIEBEN, INDIREKT und XVERWEIS kann man 
einen dynamischen Bereich aufspannen. Diese drei Funktionen kann man als Namen speichern 
(ich habe sie mal Jahr1, Jahr2 und Jahr3 genannt). 

 

Die Namen mit den Funktionen BEREICH.VERSCHIEBEN und XVERWEIS kann man 
wunderbar in einem Diagramm verwenden: 

Für XVERWEIS heißt die Formel: 

=XVERWEIS($A$18;$B$1:$G$1;$B$2:$G$13) 

Achtung: INDIREKT aber nicht! 

 

9. Fehler 

Zwar kann der Fehler #NV abgefangen werden, aber XVERWEIS kann einige Fehler 
produzieren, die entweder überprüft oder abgefangen werden sollten: 

 #WERT: Stimmen Suchmatrix und Rückgabematrix nicht in ihrer Größe überein, wird 
der Fehler #WERT zurückgegeben. Ebenso bei fehlerhaften Parameterwerten. 

 #ÜBERLAUF wie alle Array-Funktionen, die mehrere Werte zurückgeben, werden 
vorhandene Werte nicht kommentarlos überschrieben, sondern mit #ÜBERLAUF 
gekennzeichnet. 

 #NV: Wurde der Parameter wenn_nicht_gefunden nicht gefüllt, liefert die Funktion 
XVERWEIS den Fehler #NV. 

 #BEZUG: Wurde die Funktion verschoben oder kopiert, so dass der Bezug nicht 
angepasst werden kann, ist #BEZUG die Konsequenz. 

Und der Haken an der Sache? 

XVERWEIS scheint bei großen Datenmengen langsamer zu sein als SVERWEIS. Jedoch: auch 
hierfür hat XVERWEIS eine Lösung: der letzte Parameter „Suchmodus“ weist die beiden Werte 
2 und -2 auf – Binärsuche bei einer aufsteigenden und absteigenden Liste. Tatsächliche – diese 
Variante ist schneller als SVERWEIS. 

10. Zusammenfassung 

Die Vorteile von XVERWEIS 

 Fasst SVERWEIS und WVERWEIS zusammen 
 Fasst INDEX und VERGLEICH zusammen 
 Hat als Standardsuchparameter „exakte Suche“ eingestellt 
 Ist matrixfähig, das heißt: kann mehrere Werte gleichzeitig suchen – in mehreren Spalten 

und – falls nötig: mehrere Werte zurückgeben. 
 Hat die Funktion WENNNV integriert 
 Hat einen Parameter zur schnellen Suche. 

11. Die neue Funktion XVERGLEICH 

Mit dem Wissen über XVERWEIS kann nun leicht die Funktion XVERGLEICH beschrieben 
werden. Die Syntax: 

=XVERGLEICH(Suchkriterium;Suchmatrix;  
[Vergleichsmodus];[Suchmodus]) 

Die beiden Parameter Vergleichsmodus und Suchmodus haben – analog zu XVERGLEICH die 
gleichen Parameterwerte: 

Vergleichsmodus Beschreibung 
0 Genaue Übereinstimmung 
-1 Exakte Übereinstimmung oder nächst kleineres Element 
1 Exakte Übereinstimmung oder nächst größeres Element 
2 Platzhalterübereinstimmung 

 

Suchmodus Beschreibung 
1 Vom ersten zum letzten Element suchen 
-1 Vom letzten zum ersten Element suchen 
2 Binärsuche (sortiert in aufsteigender Reihenfolge) 
-2 Binärsuche (sortiert in absteigender Reihenfolge) 

 

12. Beispiel 

Damit kann man eine x-y-Suche, beispielsweise die Entfernungen von Ort zum Ort B lösen. 
Während dies einerseits die Formel 

=XVERWEIS(B2;Orte; 
XVERWEIS(C2;Entfernungen!$B$1:$AY$1;Entfernungen!$B$2:$AY$51)) 

dieses Problem löst, kann man es auch „klassisch“ lösen: 

=INDEX(Entfernungen!$B$2:$AY$51; 
XVERGLEICH(B2;Orte);XVERGLEICH(C2;Orte)) 

Dabei wurde mit „Orte“ die Spalte A2:A51 benannt – die Spalten, in denen sich die Ortsnamen 
befinden. Das Ergebnis: 

 

So kann bei einer sortierten Liste ermittelt werden, wie viele Werte sich bis zu einem bestimmten 
Wert befinden. Beispielsweise: Wie viele Kunden sind vor dem 01.01.1970 geboren? 

=XVERGLEICH(Q2;K:K;1) 

